Appendix

The model used for the example is a very simple, general linear model, easily found in most textbooks on feedback control. I have chosen to refer to the instance of this model that appears in Csete and Doyle (2002) so that interested readers can cross reference that very interesting presentation. But there is nothing particularly special about the model.

The model is shown as a traditional feedback diagram in Figure A.1. Circles represent addition or subtraction. Going around the loop starting at A: the area of cultivated land ($a$) is disturbed ($d$) by weather to produce yield ($y$). The yield is sampled and transformed into a measurement (mental model of farmer). The measurement is compared to the desired value ($r$). The cultivated land is adjusted based on the difference between desired output and a measurement of the output ($u = r - x$).

![Block diagram of simple feedback system.](https://example.com/block_diagram.png)

**Figure A.1: Block diagram of simple feedback system.**

The mathematical representation is:

\[ y = d + a \quad (1) \]
\[ u = r - x \quad (2) \]
\[ \dot{x} = k_1 y - k_2 x \quad (3) \]
\[ \dot{a} = gu \quad (4) \]

The key parameter is the gain, $g$, i.e. how fast $x$ changes in response to $u$.

**Basic ODE model for use with XPPAUT**

Interested readers may explore the model (it's fun!). You will need to download the XPPAUT package (which is available for Windows, Mac OS X, and several UNIX flavors) from the XPPAUT Home Page\(^3\).

\(^3\)http://www.math.pitt.edu/ bard/xpp/xpp.html
# Simple feedback model modified from Doyle.

par k1=0.01, k2=0.1, g=0.1, dmax=0, rmax=0.5, omega=1
par switch=1, p1=60, p2=20, hfson=50, hfoff=60

# functions
f(x,a,b)=if(x<a) then(0) else(if(x<b) then(1) else(0))

# Equations and hidden variables-----------------------------
r = rmax
d = dmax*v*(1-f(t,hfson,hfoff)) + dmax*v1*f(t,hfson,hfoff)
y = d + a
u = r - x

# differential equations-----------------------------------

# oscillators - shocks
duo/dt = uo*(1 - uo^2 - v^2) - (2*pi/p1)*v
dv/dt = v*(1 - uo^2 - v^2) + (2*pi/p1)*uo
duo1/dt = uo1*(1 - uo1^2 - v1^2) - (2*pi/p2)*v1
dv1/dt = v1*(1 - uo1^2 - v1^2) + (2*pi/p2)*uo1
init uo=-1, v=0, uo1=-1, v1=0

# feedback system
dx/dt = k1*y - k2*x
da/dt = g*u

aux yout = y
aux uout = u
aux rout = k2*r/k1
aux dist = d

@ yp=yout, total=200, xhi=200, yhi=10, maxstor=10000

done